

Massachusetts Institute of Technology
Instrumentation Laboratory
Cambridge, Massachusetts

Space Guidance Analysis Memo #16-66

TO: SGA Distribution
FROM: Raymond Morth
DATE: 18 August 1966
SUBJECT: Gimbal Lock Avoidance Schemes

Apparently there is now a surfeit of gimbal lock avoidance schemes, viz., CALC MANU and KALCMANU. Here is one more.

The main idea is to proceed with the attitude maneuver as if no problem existed and to alter the maneuver only when the gimbal lock region is entered. (Gimbal lock is defined as the region where the magnitude of the middle gimbal angle exceeds 60 degrees.) When the gimbal lock region is entered, new body rates are commanded such that the middle gimbal rate is zero and the inner gimbal rate will be a constant value such as to decrease the angle error between the desired and actual X-axis of the spacecraft. These rates are:

$$q_D = 10 \cos M \cos 0 \text{ sign (TEM MM3)}$$

$$r_D = -10 \cos M \sin 0 \text{ sign (TEM MM3)}$$

where

$$q_D = \text{desired pitch rate} \quad (\text{deg/sec})$$

$$r_D = \text{desired yaw rate} \quad (\text{deg/sec})$$

$\cos M$ = cosine middle gimbal angle

$\cos 0$, $\sin 0$ = cosine, sine outer gimbal angle

$$\text{TEM MM3} = \bar{X}_D * \bar{X}_B \cdot \bar{Y}_{PL}$$

\bar{X}_D , \bar{X}_B = desired, actual spacecraft X-axis

\bar{Y}_{PL} = Y platform axis

This will result in an inner-gimbal rate of 10 deg/sec. and a maximum rate of 5 deg/sec. in pitch and yaw.

Further if the planned maneuver will result in a decreasing middle gimbal angle, this maneuver is done.

The program in Block II basic language is shown in Fig. 1. The logic flow graph is shown in Fig. 2. The test for entering this logic is the first part of a pitch yaw autopilot shown in Fig. 3. Some small additions to this program might be required to preclude such things as singularities and logical traps.

Two worst test cases are illustrated in Fig. 4. The system response is shown in Fig. 5 and 6. No control of the outer gimbal is made in these tests.

This particular method enjoys the virtue of simplicity with the attendant features of short program length and rapid execution time. Further, an override of manual maneuvers can be made with a suitable redefinition of TEMMM3. On the negative side is the fuel economy aspect. Maneuvers with this system require three different sets of rates (four in the worst case). Another system which preplans the maneuver will establish two sets of rates (one set in the best case).

LOCKER	CCS	ALPERR	IF PLANNED MDOT DECREASES AMG, PROCEED
	CA	SINO	PLANNED MDOT = 5 SINO SIGN(BETERR)
	TC	+2	+ 5 COSO SIGN(ALPERR)
	CS	SINO	ALPERR = PITCH ERRCR=ASIN(XD*XB.ZB)
	TS	TEMMM	BETERR = YAW ERRCR = ASIN(XD*XB.YB)
	CCS	BETERR	SINC,COSO=SIN,COS AOG (SINGLE PRECISION)
	CA	COSO	
	TC	+2	
	CS	COSO	
	AD	TEMMM	
	EXTEND		
	MP	AMG	
	EXTEND		
	BZMF	A/P2	A/P2 IS NORMAL AUTOPILOT
MDCTTEST	CS	-DELAMG	IF MDOT SING(AMG) NEG, GO TO NOACTION
	MD	AMG	
	EXTEND		
	BZMF	NOACTION	
IDOTCALC	CA	SINO	COMMAND IDOT TO 10 SIGN(TEMMM3)
	EXTEND		
	MP	-10COSM	$XD*XB.YPL$ = DESIRED SIGN OF IDOT
	INDEX	TEMMM3	TEMMM3 = 30000 IF $XD*XB.YPL$ POS
	TC	A	= 40000 IF $XD*XB.YPL$ NEG
	EXTEND		YPL = YPLATFORM
	TC	TESTER	SET PITCH BITS + IF A +,0 IF A 0,- IF A -
	TC	PITCHCOD	
	CA	COSO	
	EXTEND		
	MP	-10COSM	
	INDEX	TEMMM3	
	TC	A	
	EXTEND		
	SU	YAWDOT	SET YAW BITS
	TC	TESTER	
	TC	YAWCODE	
	TC	A/P3	
TESTER	TS	TEMMM	
	CCS	A	
	AD	-RATEDB	
	TCF	+2	
	AD	-RATEDB	
	EXTEND		
	BZMF	+2	
	TCF	TEMMMTST -1	
	CA	ZERO	
	TC	Q	
NOACTION	CA	ZERO	
	TC	A/P3	A/P3 WRITES PITCH/YAW CODE IN CHANNEL

Fig. 1 Avoid Gimbal Lock Program




Fig. 2 Avoid Gimbal Lock Logic

A/P1	CA	4A1	
	TS	4A	
	CA	ZERO	
	TS	JETEM	
LOCKTEST	CS	XD1	MIDDLE COMP OF XD VECTOR
	EXTEND		IF ENDPOINT IN GIMBAL LOCK, ALARM AND
	MP	XD1	AND NO MANEUVER
	AD	1/16	
	EXTEND		
	BZMF	NOCANDO	DESIRED X AXIS IN GIMBAL LOCK --ALARM**
	CA	AMG	TEST IF AMG EXCEEDS LIMIT OF 60 DEG
	EXTEND		LIMIT IS 55 DEG WHEN IN LOCK
	MP	A	
	INDEX	TEMMM4	TEMMM4=0 INITIALLY
	AD	-LOCLMSQ	HYSTERESIS CIRCUIT FOR AMG LIMIT
	EXTEND		
	BZMF	+4	
	CA	ONF	
	TS	TEMMM4	
	TC	LOCKER	
	CA	ZERO	
	TS	TEMMM4	
A/P2	CA	ALPERR	
	TS	AA	
	CA	PITCHDOT	
	TS	RATE	
	CA	ONE	
	TS	TEMMM	
	TC	A/P	
	TC	PITCHCOD	
	CA	BETERR	
	TS	AA	
	CA	YAWDOT	
	TS	RATE	
	CA	ONF	
	TS	TEMMM	
	TC	A/P	
	TC	YAWCODE	
A/P3	EXTEND		
	WRITE	PYJETS	
	CS	PITCHDOT	UPDATE ALPERR AND BETERR
	AD	ALPERR	BY INTEGRATING PITCHDOT AND YAWDOT
	TS	ALPERR	
	CS	YAWDOT	
	AD	BETERR	
	TS	BETERR	
	TC	ENDOFJOB	
A/P	CCS	AA	
	CA	RATE	
	TCF	+6	
	AD	ONE	
	TS	AA	
	CS	ONE	
	TS	TEMMM	
	CS	RATE	
	EXTEND		
	DV	4A	

Fig. 3a Pitch-Yaw Autopilot

TS	R/4A
EXTEND	
SQUARE	
EXTEND	
MP	-SCLRDAD
AD	AA
CCS	A
AD	AA
TCF	+7D
TCF	+1
CCS	R/4A
CS	ONE
TCF	TEMMMTST
TCF	-6
TCF	-7
AD	-AMIN
CCS	A
CS	QMAX
TCF	+3
CS	-QMIN
CS	-QMIN
AD	R/4A
CCS	A
CA	ZERO
TC	Q
TCF	+1
CA	ONE
TEMMMTST	TS RATE
	CCS TEMMM
	CS RATE
	TC Q
	CA RATE
	TC Q
PITCHCOD	INDEX A
	CAF P/RJCODE
	ADS JETEM
	TC Q
YAWCODE	INDEX A
	CAF YJETCODE
	ADS JETEM
	TC Q
4A1	DEC .022222
-SCLRDAD	DEC -.444444
-AMIN	DEC -.022222
-QMIN	DEC -.05
QMAX	DEC .125
-LOCLMSQ	DEC -.111111
	DEC -.093364
1/16	DEC .0625
-RATEDB	OCT 77772
-10COSM	DEC -100

Fig. 3b Pitch-Yaw Autopilot (Cont'd)




Fig. 4a Worst Entry Case




Fig. 4b Worst Case






Fig. 5 Worst Entry Case






Fig. 6 Worst Case